

```

10 REM*****
20 REM*****
30 REM*****
40 REM*****          LOESUNGSVORSCHLAG ZU          *****
50 REM*****          TEMPERATURMESSPLATZ          *****
60 REM*****          UEBUNG a)                    *****
70 REM*****
80 REM*****          GST 04/88                    *****
90 REM*****
100 GOSUB Pca_init: REM *****
110 GOSUB Var_init
140 GOSUB Ieee_init: REM Hier werden die Unterprogramme aufgerufen !
150 GOSUB Ptm_init
160 GOSUB Ptc_init
180 GOSUB Meas_loop: REM *****
190 REM
200 END
999 REM#####ENDE DES HAUPTPROGRAMMS#####
1000Pca_init: REM=====
1010 REM          ***** PCA- Initialisierung *****
1020 PRINT "E:[2J": CLEAR
1030 RETURN
1040 REM=====
1100Var_init: REM=====
1110 REM          ***** Variablendeklaration *****
1120 Soll_ptcl=15: Grenz_ptm=24
1130 Soll_ptcu=35
1140 Hys=1: Ptm=21: Ptc=20
1150 REM
1160 RETURN
1199 REM=====
1400Ieee_init: REM=====
1410 REM          ***** Initialisierung des IEC- Bus *****
1420 IEC IFC ,DCL ,TERM 10,TIME 1000
1430 RETURN
1440 REM
1499 REM=====
1500Ptm_init: REM=====
1510 REM          ***** Initialisierung des PTM *****
1520 IEC OUT Ptm,"E1": REM ***** Schlusszeichen = LF *****
1530 IEC OUT Ptm,"C,S1": REM ***** Anzeige = Celsius, Messung ueber *****
1540 RETURN : REM          ***** Kanal 1 *****
1550 REM
1599 REM=====
1600Ptc_init: REM=====
1610 REM          ***** Initialisierung des PTC *****
1620 IEC OUT Ptc,"E1": REM ***** Schlusszeichen = LF *****
1630 IEC OUT Ptc,"S+15C": REM ***** Solltemp. bei Beginn = 15 celsius *****
1640 IEC OUT Ptc,"HU+1": REM ***** obere Hysterese 1 Grad *****
1650 IEC OUT Ptc,"HL+1": REM ***** untere Hysterese 1 Grad *****
1660 RETURN
1670 REM
1699 REM=====

```

```

1699 REM=====
2000 Meas_loop: REM=====
2010 REM *****
2020 REPEAT : REM      Abfrage des PTM, ob Aussentemperatur unter-
2025   HOLD 100: REM      schritten. Bei Ja, Einschalten des Foens
2030   IEC IN Ptm;1,Aussen$: REM      durch Setzen der PTC- Solltemperatur auf
2040   Aussen=VAL(Aussen$): REM      den Wert 35 Grad C. Bei Nein, Fortfuehren
2050 UNTIL Aussen<Grenz_ptm: REM      der Abfrage.
2060 REM *****
2070 IEC OUT Ptc,"S+35C": REM      Setzen der Solltemp. am PTC
2110 REM *****
2120 REPEAT : REM      Abfrage des PTM ob Aussentemperatur wieder
2125   HOLD 100: REM      den Sollwert nach oben ueberschritten hat.
2130   IEC IN Ptm;1,Aussen$: REM      Bei Ja, Ausschalten des Foens durch setzen
2140   Aussen=VAL(Aussen$): REM      einer niedrigen Solltemperatur.
2150 UNTIL Aussen>Grenz_ptm: REM
2160 REM *****
2170 IEC OUT Ptc,"S+15C": REM      Siehe Zeile 2070
2180 RETURN
2190 REM *****
2199 REM=====
2200 REM ** Zur Erinnerung! **
2210 REM Der Foen war an einen Lastschalter an der Rueckseite des PTC ge-
2220 REM bunden. Der Lastschalter konnte nur durch Setzen einer ueberhoeten
2230 REM oder einer zu tiefer Solltemperatur ein- und ausgeschaltet werden.
2240 REM ***
2250 REM Bedeutung der Variablen:
2260 REM Soll-ptcl ist die Groesse fuer den Ausschaltwert des PTC- Last-
2270 REM      relais (low)
2280 REM Soll-ptcu ist die Groesse fuer den Einschaltwert des PTC- Last-
2290 REM      relais (high)
2300 REM Grenz_ptm ist der simulierte Aussentemperaturwert, bei deren
2310 REM      Unterschreitung die Heizung (Foen) eingeschaltet
2320 REM      werden soll !
2330 REM*****

```